/home/m262f99/KOEPF/worksheetsV.4/hyperdemo.mws

Math 262a, Fall 1999, Glenn Tesler
Polynomial/Hypergeometric solutions of recurrences
11/8/99

```
> read `hsum.mpl`;
```

Copyright 1998 Wolfram Koepf, Konrad-Zuse-Zentrum Berlin

Example 1. Polynomial solutions of a recurrence equation.
Since (n*(n+1)*(n+2,8))^2 - (n,8)*(n+8)*(n+9))^2 = 0, we must have

f(n) = pochhammer(n,8)^2 = (n(n+1)...(n+7))^2

as a solution of

(n(n+1))^2 f(n+2) - ((n+8)(n+9))^2 f(n) = 0

```
> recpoly((n*(n+1))^2*f(n+2) - ((n+8)*(n+9))^2*f(n),
  f(n));
```

$$\delta_0 \, n^2 (n+7)^2 (n+6)^2 (n+5)^2 (n+4)^2 (n+3)^2 (n+2)^2 (n+1)^2$$

delta_0 is an arbitrary constant.

Example 2. Hypergeometric solutions of a recurrence equation.

```
> f1 := hyperterm([a,b],[c],x,n);
```

$$f1 := \frac{\mathrm{pochhammer}(a, n)\, \mathrm{pochhammer}(b, n)\, x^n}{\mathrm{pochhammer}(c, n)\, n!}$$

```
> simpcomb(subs(n=n+2,f1)/f1);
```

$$\frac{(a+n)(a+n+1)(b+n)(b+n+1)\, x^2}{(c+n)(c+n+1)(n+1)(n+2)}$$

```
> eq := numer(")*f(n) - denom(")*f(n+2);
```

$$eq := (a+n)(a+n+1)(b+n)(b+n+1)\, x^2\, \mathrm{f}(n) - (c+n)(c+n+1)(n+1)(n+2)\, \mathrm{f}(n+2)$$

By construction, this is solved by the hypergeometric term

2F1(a,b;c;x)_n

```
> ratios := rechyper(eq,f(n));
```

$$ratios := \{-\frac{x\,(b+n)(a+n)}{(n+1)(c+n)}, \frac{x\,(b+n)(a+n)}{(n+1)(c+n)}\}$$

It doesn't return f(n), but rather the ratio f(n+1)/f(n). Also, it found a SECOND hypergeometric solution:

2F1(a,b;c; -x)_n

which in retrospect isn't surprising.

```
> ratiosk := subs(n=k,ratios):
> product(ratiosk[1],k=0..n-1);
  product(ratiosk[2],k=0..n-1);
```

$$\frac{x^n\,\Gamma(b+n)\,\Gamma(a+n)\,\Gamma(c)}{\Gamma(n+1)\,\Gamma(c+n)\,\Gamma(b)\,\Gamma(a)}$$

$$\frac{(-1)^n\,x^n\,\Gamma(b+n)\,\Gamma(a+n)\,\Gamma(c)}{\Gamma(n+1)\,\Gamma(c+n)\,\Gamma(b)\,\Gamma(a)}$$

Application to factorization of operators

Given a root x0 in C of a polynomial f(x) over Q, then f(x)=g(x)*(x-x0) for some polynomial g(x) over C.
If the minimal polynomial of x0 over Q is h(x), then f(x)=p(x)h(x) for some polynomial p(x) over Q.

Now consider a differential operator L = sum a_i(x) Dx^i, each a_i(x) in Q(x), that annihilates a function f(x), or a recurrence operator that annihilates a function f(n), etc.

The first order operator R = f(x) Dx - f'(x) annihilates f(x) in the differential case, and R = f(n) En - f(n+1) annihilates f(n) in the shift case. f is the analogue of a root x0 in an extension field; R is the analogue of the linear factor x-x0. Using non-commutative right division of L by R we obtain a factorization L = G R, with G, R operators (over the extension, usually not over Q(x)).

There is a minimal order operator H over Q(x) that annihilates f. Using non-commutative right division of L by H gives an operator P over Q(x) with L = P H.

There is software available for these divisions that we will use later. Non-commutative factorization in general is only a partially solved problem, and I don't have anyone's software for it.

```
> L := subs(f(n)=1,f(n+1)=En,f(n+2)=En^2,eq);
```

$$L := (a+n)\,(a+n+1)\,(b+n)\,(b+n+1)\,x^2-(c+n)\,(c+n+1)\,(n+1)\,(n+2)\,En^2$$

```
> R_op := En - ratios[1]; Rf := f(n+1) - ratios[1]*f(n);
```

$$R\_op := En+\frac{x\,(b+n)\,(a+n)}{(n+1)\,(c+n)}$$

$$Rf := \mathrm{f}(n+1)+\frac{x\,(b+n)\,(a+n)\,\mathrm{f}(n)}{(n+1)\,(c+n)}$$

```
> G_op := A*En - B; GRf := A*subs(n=n+1,Rf) - B*Rf;
```

$$G\_op := A\,En - B$$

$$GRf := A\left(\mathrm{f}(n+2) + \frac{x\,(b+n+1)\,(a+n+1)\,\mathrm{f}(n+1)}{(n+2)\,(c+n+1)}\right) - B\left(\mathrm{f}(n+1) + \frac{x\,(b+n)\,(a+n)\,\mathrm{f}(n)}{(n+1)\,(c+n)}\right)$$

```
> fs := {f(n),f(n+1),f(n+2)}: collect(GRf - eq, fs):
  coeffs(",fs):
  sols := factor(solve({"},{A,B}));
```

$$sols := \{A = -(c+n)\,(c+n+1)\,(n+1)\,(n+2),\; B = -(a+n+1)\,(b+n+1)\,x\,(n+1)\,(c+n)\}$$

```
> map(collect,factor(subs(sols,G_op)),En,factor);
```

$$(n+1)\,(c+n)\,(-(n+2)\,(c+n+1)\,En + x\,(b+n+1)\,(a+n+1))$$

The factored form of L obtained by using the first ratio to generate a right factor is

```
> factorL := " * R_op;
```

$$factorL := (n+1)\,(c+n)\,(-(n+2)\,(c+n+1)\,En + x\,(b+n+1)\,(a+n+1))\left(En + \frac{x\,(b+n)\,(a+n)}{(n+1)\,(c+n)}\right)$$

Example 3. Failure to find a hypergeometric solution.

Now test the recurrence obtained on my handout "Creative telescoping for a triple integral":

The sequence a(n) = binomial(n,n/2)^2 satisfies a recurrence:

```
> rechyper(16*(n+1)^2*a(n) - (n+2)^2*a(n+2), a(n));
```

$$\{\ \}$$

There’s no hypergeometric solutions because binomial(n,n/2) is 2-fold hypergeometric, not 1-fold. This can be detected automatically, but not with this software. (Even that can fail if there’s no k-fold solutions for any k, which is possible.) Substitute n=2m, b(m)=a(n/2):

```
> rechyper(16*(2*m+1)^2 * b(m) - (2*m+2)^2 * b(m+1),
  b(m));
```

$$\{4\,\frac{(2\,m+1)^2}{(m+1)^2}\}$$

```
> subs(m=m0,"[1]): product(",m0=0..m-1);
```

$$\frac{4^m\,(2^m)^2\,\Gamma\left(m+\frac{1}{2}\right)^2}{\Gamma(m+1)^2\,\pi}$$

```
> an1 := simplify(subs(m=n/2,"));
```

$$an1 := \frac{2^{(2\,n)}\,\Gamma\left(\frac{1}{2}\,n+\frac{1}{2}\right)^2}{\Gamma\left(\frac{1}{2}\,n+1\right)^2\,\pi}$$

This is the same as binomial(n,n/2)^2 (identifying binomial coefficients with their def. in terms of factorials, and factorials in terms of Gamma functions, so that non-nonnegative integer arguments are valid).

```
> simplify( an1 / binomial(n,n/2)^2);
```

$$1$$

```
>
```